
PROC DS2, the VDW, and You

What, When, Why?

Originally created as PROC TSPL  (Table Server Programming Language), a specialized procedure for database programming

Renamed to DS2  (DATA Step Two) and pitched as a modernized version of the classic DATA  step

Released as an "experimental" procedure in SAS 9.3M2 and a production procedure in SAS 9.4

A Brief Overview of DS2 Features

PROC DS2  is a procedure for advanced data management & manipulation.

Most classic DATA  step features are maintained in some cases with better performance

New features in DS2  include

Fast multithreaded execution of CPU-intensive operations

SQL integration

Support for a wider variety of data types

More verbose, less ambiguous syntax

Support for "object-oriented" programming

Should You be Using DS2?

Agenda

1. A brief overview of DS2  syntax, with comparison to the DATA  step

2. General use cases for PROC DS2

3. The case for DS2  in VDW workplans and QAs

Basics

If you have a DATA  step that looks like this:

DATA out;
    LENGTH <variable definitions>;
    SET in;
    <additional statments>
    DROP <variable list>;
RUN;

The equivalent DS2  code looks like this:

PROC DS2;
    DATA out / overwrite=yes;
        DECLARE <variable definitions>;
        DROP <variable list>;
        METHOD run();
            SET in;
            <additional statments>
        END;
    RUN;
QUIT;

Declarative statments can be in global or local scope.

Executable statements go inside a METHOD  block.

All else equal, writing DS2  is more work, but the syntax is less ambiguous than the DATA  step and allows for variable scoping.

Faster performance on wide data

8          ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
8        ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
9          
10         data NLP_covariates;
11             array covariates {*} &cuis;
12             set mml_full_nlp_afep2;
13             by patient_ID;
14             retain n;
15         
16             if _n_ EQ 1 then n = dim(covariates);
17         
18             if first.patient_ID then
19                 do i = 1 to n;
20                     covariates{i} = 0;
21                 end;
22         
23             do i = 1 to n;
24                 covariates{i} + (cui EQ vname(covariates{i}));
25             end;
26         
27             if last.patient_ID then output;
28             drop doc_ID CUI negated n i;
29         run;

NOTE: There were 1203319 observations read from the data set WORK.MML_FULL_NLP_AFEP2.
NOTE: The data set WORK.NLP_COVARIATES has 1844 observations and 168 variables.
NOTE: DATA statement used (Total process time):
      real time           4:08.71
      cpu time            4:06.42
      

30         
31         
32         ods html5 (id=saspy_internal) close;ods listing;
33         

34         

Array operations seem to be faster in DS2

37         ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
37       ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
38         
39         proc ds2 libs=work;
40             data nlp_covariates(drop=(doc_id cui negated n i)) / overwrite=yes;
41                 declare int n i;
42                 retain n;
43                 vararray int covariates[*] &cuis;
44         
45                 method init();
46                     n = dim(covariates);
47                 end;
48         
49                 method run();
50                     set mml_full_nlp_afep2;
51                     by patient_id;
52                     if first.patient_id then do i = 1 to n;
53                         covariates[i] = 0;
54                     end;
55                     do i = 1 to n;
56                         covariates[i] + (cui EQ upcase(vname(covariates[i])));
57                     end;
58                     if last.patient_id then output ;
59                 end;
60             run;
NOTE: BASE driver, creation of a INTEGER column has been requested, but is not supported by the BASE driver. A DOUBLE PRECISION 
      column has been created instead.
NOTE: Execution succeeded. 1844 rows affected.
61         quit;

NOTE: PROCEDURE DS2 used (Total process time):
      real time           33.13 seconds
      cpu time            34.37 seconds
      

62         
63         
64         ods html5 (id=saspy_internal) close;ods listing;
65         

66         

And the multithreaded version is even faster

69         ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
69       ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
70         
71         proc ds2 libs=work;
72             thread nlp_covariates_thread / overwrite=yes;
73                 declare int n i;
74                 retain n;
75                 vararray int covariates[*] &cuis;
76         
77                 method init();
78                     n = dim(covariates);
79                 end;
80         
81                 method run();
82                     set mml_full_nlp_afep2;
83                     by patient_id;
84                     if first.patient_id then do i = 1 to n;
85                         covariates[i] = 0;
86                     end;
87                     do i = 1 to n;
88                         covariates[i] + (cui EQ upcase(vname(covariates[i])));
89                     end;
90                     if last.patient_id then output ;
91                 end;
92             endthread;
93         
94             data nlp_covariates(drop=(doc_id cui negated n i)) / overwrite=yes;
95                 declare thread nlp_covariates_thread t;
96                 method run();
97                     set from t threads=4;
98                     by patient_ID;
99                 end;
100            run;
NOTE: Created thread nlp_covariates_thread in data set work.nlp_covariates_thread.
NOTE: BASE driver, creation of a INTEGER column has been requested, but is not supported by the BASE driver. A DOUBLE PRECISION 
      column has been created instead.
NOTE: Execution succeeded. 1844 rows affected.
101        quit;

NOTE: PROCEDURE DS2 used (Total process time):
      real time           16.48 seconds
      cpu time            1:04.56
      

102        
103        
104        ods html5 (id=saspy_internal) close;ods listing;
105        

106        

Data types

DS2  supports a wide variety of data types for better compatibility with external DBMSs | Classic SAS data types | DS2 data types | |------------------------|-------------------------------------------------------------------------------------------

-------------------------| | Character | CHAR(n), NCHAR(n), NVARCHAR(n), VARCHAR(n) | | Numeric | BIGINT, BINARY(n), DATE, DECIMAL(p, s), DOUBLE, FLOAT, INTEGER, REAL, TIME(p), TIMESTAMP(p), TINYINT,

VARBINARY(n) | | | |

When writing back to a base SAS dataset (I.E. .SAS7BDAT), everything still gets converted back to regular SAS Character and Numeric with appropriate formats.

The DS2  DECLARE  statement is more flexible than the LENGTH  statement in a DATA  step.

Suppose we have two datasets with conflicting data types

Date read as character

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat

2 charvar Char 1 $1. $1.

3 datevar Char 1 $1. $1.

1 id Num 8 BEST12. BEST32.

4 numvar Char 1 $1. $1.

Date read as numeric

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat

2 charvar Char 5 $5. $5.

3 datevar Num 8 YYMMDD10. YYMMDD10.

1 id Num 8 BEST12. BEST32.

4 numvar Num 8 BEST12. BEST32.

This causes an error in the DATA  step

126        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
126      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
127        
128        data all;
129            set one two;
ERROR: Variable datevar has been defined as both character and numeric.
ERROR: Variable numvar has been defined as both character and numeric.
130        run;

NOTE: The SAS System stopped processing this step because of errors.
WARNING: The data set WORK.ALL may be incomplete.  When this step was stopped there were 0 observations and 4 variables.
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.01 seconds
      

131        
132        
133        ods html5 (id=saspy_internal) close;ods listing;
134        

135        

ERROR: Variable datevar has been defined as both character and numeric. None

We can't fix it with a LENGTH  statement either

138        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
138      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
139        
140        data fix_data_types;
141            length charvar $ 4 datevar numvar 8;
142            set one;
ERROR: Variable datevar has been defined as both character and numeric.
ERROR: Variable numvar has been defined as both character and numeric.
143        run;

NOTE: The SAS System stopped processing this step because of errors.
WARNING: The data set WORK.FIX_DATA_TYPES may be incomplete.  When this step was stopped there were 0 observations and 4 variables.
NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.01 seconds
      

144        
145        
146        ods html5 (id=saspy_internal) close;ods listing;
147        

148        

ERROR: Variable datevar has been defined as both character and numeric. None

But we can enforce the data types we want with a DECLARE  in DS2

Types enforced by DS2

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat Label

1 charvar Char 4    

3 datevar Num 8 DATE9.   

4 id Num 8 BEST12. BEST32. id

2 numvar Num 8    

SQL integration

175        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
175      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
176        
177        %setup(mssql)
NOTE: Libref MSSQLVDW was successfully assigned as follows: 
      Engine:        ODBC 
      Physical Name: 

NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.00 seconds
      

205        
206        proc sql;
207            create table demo as
208                select
209                     a.mrn, a.birth_date, a.race1, a.race2, a.race3, a.race4, a.race5, a.sex_admin
210                from
211                    &_vdw_demographic a
212                left join
213                    &_vdw_death b
214                    on a.mrn = b.mrn
215                left join
216                    &_cesr_exclusion c
217                    on a.mrn = c.mrn
218                where
219                    a.birth_date >= '01JAN2000'd and
220                    b.mrn is NULL and
221                    c.mrn is NULL;
NOTE: Table WORK.DEMO created, with 1785 rows and 8 columns.

222        quit;
NOTE: PROCEDURE SQL used (Total process time):
      real time           0.14 seconds
      cpu time            0.01 seconds
      

223        
224        
225        ods html5 (id=saspy_internal) close;ods listing;
226        

227        

SELECT  inside of SET

230        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
230      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
231        
232        %setup(mssql)
NOTE: Libref MSSQLVDW was successfully assigned as follows: 
      Engine:        ODBC 
      Physical Name: 

NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.01 seconds
      

260        
261        proc ds2;
WARNING: WARNING: [Microsoft][ODBC Driver 13 for SQL Server][SQL Server]Changed database context to 'WORK7'.
WARNING: WARNING: [Microsoft][ODBC Driver 13 for SQL Server][SQL Server]Changed language setting to us_english.
WARNING: The license for SAS product ID 'ODBC' will be expiring soon and is currently in warning mode to indicate this upcoming 
         expiration.  Most typically this warning period runs for 45 days. Please run PROC SETINIT to obtain more information on 
         your warning period.
262            data demo / overwrite=yes;
263                method run();
264                    set {
265                        select
266                             a.mrn, a.birth_date, a.race1, a.race2, a.race3, a.race4, a.race5, a.sex_admin
267                        from
268                            &_vdw_demographic a
269                        left join
270                            &_vdw_death b
271                            on a.mrn = b.mrn
272                        left join
273                            &_cesr_exclusion c
274                            on a.mrn = c.mrn
275                        where
276                            a.birth_date >= date '2000-01-01' and
277                            b.mrn is NULL and
278                            c.mrn is NULL
279                    };
280                end;
281            run;
NOTE: BASE driver, creation of a VARCHAR column has been requested, but is not supported by the BASE driver. A CHAR column has been 
      created instead.
NOTE: BASE driver, creation of a DATE column has been requested, but is not supported by the BASE driver. A DOUBLE PRECISION column 
      has been created instead. A format has been associated with each column.
NOTE: Execution succeeded. 1785 rows affected.
282        quit;

NOTE: PROCEDURE DS2 used (Total process time):
      real time           0.50 seconds
      cpu time            0.21 seconds
      

283        
284        
285        ods html5 (id=saspy_internal) close;ods listing;
286        

287        

DS2 , the VDW, and the Oracle problem

The VDW exists to facilitate the uniform creation of analytic datasets across sites

Sites should be allowed to use the DBMS platform and ETL tools of their choice, as long as we can query it from SAS.

The lead site programmer should not need to know the details of the computing environments at the contributing sites.

Distributed code should aim to require as little site-specific customization as possible, while being reasonably efficient.

Storytime:

I sent out a workplan with this:

proc sql;
    create table hfdx as
        select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
        from &_vdw_dx a
        inner join &_vdw_utilization b
            on a.mrn = b.mrn and a.enc_id = b.enc_id
            and (
                   (dx_codetype = '09' and dx in(&icd9hfdx ))
                or (dx_codetype = '10' and dx in(&icd10hfdx))
                )
            and a.adate between '01JUL2006'd and '01JAN2021'd
            and a.enctype in('IP','AV','ED')
        order by mrn, adate, ddate;
quit;

...and when I get back the logs, I see the site programmer has had to do this:

PROC SQL NOERRORSTOP;
         CONNECT TO oracle (user="XXXXXXXX" password=XXXXX path=XXXXXX);

        EXECUTE
             (CREATE TABLE XXXXXXX.XXXXXX_DX AS

                   (
                        select a.*, b.ddate

                        from &_vdw_dx  a
                             inner join &_vdw_utilization b
                                   on  a.mrn = b.mrn and a.enc_id = b.enc_id
            and (
               (dx_codetype = '09' and dx in(&icd9hfdx ))
            or (dx_codetype = '10' and dx in(&icd10hfdx))
            )
            and trunc(a.adate) between to_date('07/01/2006','mm/dd/yyyy')
            and to_date('12/30/2019','mm/dd/yyyy')
            and a.enctype in('IP','AV','ED')
                           )

       ) BY oracle;

     DISCONNECT FROM oracle;
 QUIT;

data hfdx;
    set   XXXXXXX.XXXXXX_DX(dbsastype=(  adate='date9.' Ddate='date9.'));
run;

What Went Wrong?

290        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
290      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
291        
292        %setup(mssql)
NOTE: Libref MSSQLVDW was successfully assigned as follows: 
      Engine:        ODBC 
      Physical Name: 

NOTE: DATA statement used (Total process time):
      real time           0.00 seconds
      cpu time            0.00 seconds
      

320        
321        proc sql;
322            create table hfdx_mssql as
323                select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
324                from &_vdw_dx a
325                inner join &_vdw_utilization b
326                    on a.mrn = b.mrn and a.enc_id = b.enc_id
327                    and (
328                           (dx_codetype = '09' and dx in(&icd9hfdx ))
329                        or (dx_codetype = '10' and dx in(&icd10hfdx))
330                        )
331                    and a.adate between '01JUL2006'd and '01JAN2021'd
332                    and a.enctype in('IP','AV','ED')
333                order by mrn, adate, ddate;
NOTE: Table WORK.HFDX_MSSQL created, with 2370 rows and 7 columns.

334        quit;
NOTE: PROCEDURE SQL used (Total process time):
      real time           0.15 seconds
      cpu time            0.07 seconds
      

335        
336        
337        ods html5 (id=saspy_internal) close;ods listing;
338        

339        

Let's Try it in Oracle

342        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
342      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
343        
344        %setup(oracle)
NOTE: Libref MSSQLVDW has been deassigned.
NOTE: The product with which the engine ORACLE is associated will be expiring soon, and is currently in warning mode to indicate 
      this upcoming expiration.  Please run PROC SETINIT to obtain more information on your warning period.
NOTE: Libref ORAVDW was successfully assigned as follows: 
      Engine:        ORACLE 
      Physical Name: PRODNWM_USR
372        
373        proc sql;
374            create table hfdx_ora as
375                select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
376                from &_vdw_dx a
377                inner join &_vdw_utilization b
378                    on a.mrn = b.mrn and a.enc_id = b.enc_id
379                    and (
380                           (dx_codetype = '09' and dx in(&icd9hfdx ))
381                        or (dx_codetype = '10' and dx in(&icd10hfdx))
382                        )
383                    and a.adate between '01JUL2006'd and '01JAN2021'd
384                    and a.enctype in('IP','AV','ED')
385                order by mrn, adate, ddate;
NOTE: Table WORK.HFDX_ORA created, with 2370 rows and 7 columns.

386        quit;
NOTE: PROCEDURE SQL used (Total process time):
      real time           6.95 seconds
      cpu time            0.12 seconds
      

387        
388        
389        ods html5 (id=saspy_internal) close;ods listing;
390        

391        

Hold on...

Dates from MSSQL...

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat Label

2 ENC_ID Char 64 $64. $64. ENC_ID

1 MRN Char 64 $64. $64. MRN

6 adate Num 8 DATE9. DATE9. adate

7 ddate Num 8 DATE9. DATE9. ddate

4 dx Char 20 $20. $20. dx

5 dx_codetype Char 2 $2. $2. dx_codetype

3 enctype Char 2 $2. $2. enctype

...but datetimes from Oracle

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat Label

6 ADATE Num 8 DATETIME20. DATETIME20. ADATE

7 DDATE Num 8 DATETIME20. DATETIME20. DDATE

4 DX Char 20 $20. $20. DX

5 DX_CODETYPE Char 2 $2. $2. DX_CODETYPE

3 ENCTYPE Char 2 $2. $2. ENCTYPE

2 ENC_ID Char 64 $64. $64. ENC_ID

1 MRN Char 64 $64. $64. MRN

Why not use dbsastype ?

It works, but it breaks implicit passthrough. That's why our site programmer had to do this:

PROC SQL NOERRORSTOP;
         CONNECT TO oracle (user="XXXXXXXX" password=XXXXX path=XXXXXX);

        EXECUTE
             (CREATE TABLE XXXXXXX.XXXXXX_DX AS

                   (
                        select a.*, b.ddate

                        from &_vdw_dx  a
                             inner join &_vdw_utilization b
                                   on  a.mrn = b.mrn and a.enc_id = b.enc_id
            and (
               (dx_codetype = '09' and dx in(&icd9hfdx ))
            or (dx_codetype = '10' and dx in(&icd10hfdx))
            )
            and trunc(a.adate) between to_date('07/01/2006','mm/dd/yyyy')
            and to_date('12/30/2019','mm/dd/yyyy')
            and a.enctype in('IP','AV','ED')
                           )

       ) BY oracle;

     DISCONNECT FROM oracle;
 QUIT;

data hfdx;
    set   XXXXXXX.XXXXXX_DX(dbsastype=(  adate='date9.' Ddate='date9.'));
run;

The DS2 Solution

Embedded SQL can achieve implicit passthrough.

By using declare  we can enforce the types we want on the SAS side.

This step runs on MSSQL, Oracle, and SAS datasets without any customization.

411        ods listing close;ods html5 (id=saspy_internal) file=_tomods1 options(bitmap_mode='inline') device=svg style=HTMLBlue;
411      ! ods graphics on / outputfmt=png;
NOTE: Writing HTML5(SASPY_INTERNAL) Body file: _TOMODS1
412        
413        %setup(oracle)
NOTE: Libref ORAVDW was successfully assigned as follows: 
      Engine:        ORACLE 
      Physical Name: PRODNWM_USR
441        
442        proc ds2;
443            data hfdx_ora2 / overwrite=yes;
444                declare date adate ddate;
445                method run();
446                    set {
447                        select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
448                        from &_vdw_dx a
449                        inner join &_vdw_utilization b
450                            on a.mrn = b.mrn and a.enc_id = b.enc_id
451                            and (
452                                   (dx_codetype = '09' and dx in(&icd9hfdx ))
453                                or (dx_codetype = '10' and dx in(&icd10hfdx))
454                                )
455                            and a.adate between date '2006-07-01' and date '2021-07-01'
456                            and a.enctype in('IP','AV','ED')
457                    };
458                    by mrn adate ddate;
459                end;
460            run;
NOTE: BASE driver, fixed byte UNICODE encodings not supported, using UTF-8 instead
WARNING: BASE driver, table encoding reset to utf-8
NOTE: BASE driver, creation of a VARCHAR column has been requested, but is not supported by the BASE driver. A CHAR column has been 
      created instead.
NOTE: BASE driver, creation of a DATE column has been requested, but is not supported by the BASE driver. A DOUBLE PRECISION column 
      has been created instead. A format has been associated with each column.
NOTE: Execution succeeded. 2429 rows affected.
461        quit;

NOTE: PROCEDURE DS2 used (Total process time):
      real time           2.86 seconds
      cpu time            0.37 seconds
      

462        
463        
464        
465        ods html5 (id=saspy_internal) close;ods listing;
466        

467        

It works!

Dates from MSSQL...

Alphabetic List of Variables and Attributes

# Variable Type Len Format Informat Label

2 ENC_ID Char 64 $64. $64. ENC_ID

1 MRN Char 64 $64. $64. MRN

6 adate Num 8 DATE9. DATE9. adate

7 ddate Num 8 DATE9. DATE9. ddate

4 dx Char 20 $20. $20. dx

5 dx_codetype Char 2 $2. $2. dx_codetype

3 enctype Char 2 $2. $2. enctype

...and Oracle too!

Alphabetic List of Variables and Attributes

# Variable Type Len Format Label

6 DX Char 80  DX

7 DX_CODETYPE Char 8  DX_CODETYPE

5 ENCTYPE Char 8  ENCTYPE

4 ENC_ID Char 256  ENC_ID

3 MRN Char 256  MRN

1 adate Num 8 DATE9.  

2 ddate Num 8 DATE9.  

If you read through this entire notebook, thank you!

If you have any questions or would like me to consult on any DS2 related topics please feel free to contact me at Matthew.T.Slaughter@kpchr.org.

In [2]: data NLP_covariates;
    array covariates {*} &cuis;
    set mml_full_nlp_afep2;
    by patient_ID;
    retain n;

    if _n_ EQ 1 then n = dim(covariates);

    if first.patient_ID then
        do i = 1 to n;
            covariates{i} = 0;
        end;

    do i = 1 to n;
        covariates{i} + (cui EQ vname(covariates{i}));
    end;

    if last.patient_ID then output;
    drop doc_ID CUI negated n i;
run;

In [3]: proc ds2 libs=work;
    data nlp_covariates(drop=(doc_id cui negated n i)) / overwrite=yes;
        declare int n i;
        retain n;
        vararray int covariates[*] &cuis;
        
        method init();
            n = dim(covariates);
        end;        
        
        method run();
            set mml_full_nlp_afep2;
            by patient_id;
            if first.patient_id then do i = 1 to n;
                covariates[i] = 0;
            end;
            do i = 1 to n;
                covariates[i] + (cui EQ upcase(vname(covariates[i])));
            end;
            if last.patient_id then output ;
        end;
    run;
quit;

In [4]: proc ds2 libs=work;
    thread nlp_covariates_thread / overwrite=yes;
        declare int n i;
        retain n;
        vararray int covariates[*] &cuis;

        method init();
            n = dim(covariates);
        end;

        method run();
            set mml_full_nlp_afep2;
            by patient_id;
            if first.patient_id then do i = 1 to n;
                covariates[i] = 0;
            end;
            do i = 1 to n;
                covariates[i] + (cui EQ upcase(vname(covariates[i])));
            end;
            if last.patient_id then output ;
        end;
    endthread;

    data nlp_covariates(drop=(doc_id cui negated n i)) / overwrite=yes;
        declare thread nlp_covariates_thread t;
        method run();
            set from t threads=4;
            by patient_ID;
        end;
    run;
quit;

In [5]: ods select variables (persist);
title 'Date read as character';
proc contents data=one;
run;
title 'Date read as numeric';
proc contents data=two;
run;
ods select all;

In [6]: data all;
    set one two;
run;

In [7]: data fix_data_types;
    length charvar $ 4 datevar numvar 8;
    set one;
run;

In [8]: proc ds2;
    data fix_data_types / overwrite=yes;
        declare char(4) charvar;
        declare double numvar;
        declare double datevar having format date9.;
        method run();
            set one;
        end;
    run;
quit;

title 'Types enforced by DS2';
ods select variables;
proc contents data=fix_data_types;
run;

In [9]: %setup(mssql)

proc sql;
    create table demo as
        select
             a.mrn, a.birth_date, a.race1, a.race2, a.race3, a.race4, a.race5, a.sex_admin
        from
            &_vdw_demographic a
        left join
            &_vdw_death b
            on a.mrn = b.mrn
        left join
            &_cesr_exclusion c
            on a.mrn = c.mrn
        where
            a.birth_date >= '01JAN2000'd and
            b.mrn is NULL and
            c.mrn is NULL;
quit;

In [10]: %setup(mssql)

proc ds2;
    data demo / overwrite=yes;
        method run();
            set {
                select
                     a.mrn, a.birth_date, a.race1, a.race2, a.race3, a.race4, a.race5, a.sex_admin
                from
                    &_vdw_demographic a
                left join
                    &_vdw_death b
                    on a.mrn = b.mrn
                left join
                    &_cesr_exclusion c
                    on a.mrn = c.mrn
                where
                    a.birth_date >= date '2000-01-01' and
                    b.mrn is NULL and
                    c.mrn is NULL
            };
        end;
    run;
quit;

In [11]: %setup(mssql)

proc sql;
    create table hfdx_mssql as
        select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
        from &_vdw_dx a
        inner join &_vdw_utilization b
            on a.mrn = b.mrn and a.enc_id = b.enc_id
            and (
                   (dx_codetype = '09' and dx in(&icd9hfdx ))
                or (dx_codetype = '10' and dx in(&icd10hfdx))
                )
            and a.adate between '01JUL2006'd and '01JAN2021'd
            and a.enctype in('IP','AV','ED')
        order by mrn, adate, ddate;
quit;

In [12]: %setup(oracle)

proc sql;
    create table hfdx_ora as
        select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
        from &_vdw_dx a
        inner join &_vdw_utilization b
            on a.mrn = b.mrn and a.enc_id = b.enc_id
            and (
                   (dx_codetype = '09' and dx in(&icd9hfdx ))
                or (dx_codetype = '10' and dx in(&icd10hfdx))
                )
            and a.adate between '01JUL2006'd and '01JAN2021'd
            and a.enctype in('IP','AV','ED')
        order by mrn, adate, ddate;
quit;

In [13]: ods select variables(persist);
title 'Dates from MSSQL...';
proc contents data=hfdx_mssql;
run;
title '...but datetimes from Oracle';
proc contents data=hfdx_ora;
run;
ods select all;

In [14]: %setup(oracle)

proc ds2;
    data hfdx_ora2 / overwrite=yes;
        declare date adate ddate;
        method run();
            set {
                select a.mrn, a.enc_id, a.enctype, a.dx, a.dx_codetype, a.adate, b.ddate
                from &_vdw_dx a
                inner join &_vdw_utilization b
                    on a.mrn = b.mrn and a.enc_id = b.enc_id
                    and (
                           (dx_codetype = '09' and dx in(&icd9hfdx ))
                        or (dx_codetype = '10' and dx in(&icd10hfdx))
                        )
                    and a.adate between date '2006-07-01' and date '2021-07-01'
                    and a.enctype in('IP','AV','ED')
            };
            by mrn adate ddate;
        end;
    run;
quit;

In [15]: ods select variables(persist);
title 'Dates from MSSQL...';
proc contents data=hfdx_mssql;
run;
title '...and Oracle too!';
proc contents data=hfdx_ora2;
run;
ods select all;


